
Abstract We consider the design of a circuit description library for a pure functional lan-
guage where circuits are defined as functions and connected together by writing applicative
expressions. Just like the return value of any other function in a functional program, the
output of a circuit can be bound to a variable and referred to many times. Usually it is the
programmer’s intention that such references represent sharing of a circuit’s output, implying
a fanout structure. However, in a pure functional language, references are transparent and
the circuit description library can only view finite graph-shaped circuit structures as infi-
nitely expanded regular trees. To overcome this problem, we introduce expressible sharing,
a technique in which the programmer expresses the fork-points present in a circuit in the
same way that they express other circuit components like and-gates and or-gates. We define
a library for circuit description which uses expressible sharing and demonstrate it on a small
case-study.

Keywords Functional programming · Circuit description · Data structures · Sharing

1 Introduction

In a pure functional program we are free to substitute any occurrence of a function defini-
tion’s left-hand side with its right-hand side (and vice-versa), while preserving the behav-
iour of the program, subject to restrictions on totality and pattern-matching. This attractive
property is known as referential transparency, and one consequence of it is that cyclic data
structures become indistinguishable from an infinitely expanded regular trees. To illustrate,
consider the definition of an infinite list of ones in the pure functional language Haskell.

ones :: [Int]
ones = 1:ones

Fig. 1 The data structure ones
represented as a graph (a) and as
an infinite tree (b), and the data
structure 1:ones represented as
a graph (c)

Fig. 2 Circuit diagram for flipper. The large triangle represents a D-type flip-flop whose first input is
emitted on the first clock-cycle, and whose second input, delayed by one clock-cycle, is emitted thereafter

Although ones has a natural interpretation as a graph (Fig. 1(a)), and may even be imple-
mented as such, it can only be viewed as an infinite tree (Fig. 1(b)) by a Haskell program.
If a program could view ones as a graph, then it could distinguish ones (Fig. 1(a)) from
1:ones (Fig. 1(c)). Hence the program’s result could change after replacing equals for
equals, and referential transparency would be lost.

In this paper we consider an application in which we would like to be able to observe
cycles, or more generally sharing, in a data structure. Our application is a Haskell library
for describing digital circuits. Haskell turns out to be very suitable in this role, as surveyed
by Sheeran in [6], but the loss of sharing information when generating a data structure
representing a circuit is problematic. To illustrate, consider the following description of the
circuit flipper shown in Fig. 2.

flipper :: Bit -> Bit
flipper init = out

where out = inv (flipFlop init out)

In the definition of flipper, if we were to define the components flipFlop and inv
to be the data constructors FF and Inv respectively, then evaluation of flipper would
yield an infinitely large data structure, in much the same way that ones does. In our appli-
cation this is a problem, because we would like to determine a graph-shaped data structure,
representative of a circuit, which could be used, for example, to realise the described circuit.

We investigate a technique, which we call expressible sharing, as a solution to this prob-
lem. The basic idea is this: express the fork-points in a circuit in the same way that other
primitive components like flipFlop and inv are expressed. In the case of flipper, we
simply need to express the fork-point represented by the black dot in its diagram. The main
attractions of expressible sharing are:

• It is simple to use, requiring that the fork-points in the circuit are treated as components.
• It allows missing fork-points to be detected. If programmers omit fork-points in their

descriptions then they can be informed of this (Sect. 3.5).

On the other hand, the main drawback is:

• The programmer cannot refer to directly to constant values high and low. Instead, these
must be treated as inputs to the circuit (Sect. 3.2).

This paper is structured as follows. Section 2 presents a small Haskell library for descrip-
tion and simulation of digital circuits, and explains the problem that arises when we try to
extract netlists from such descriptions. Section 3 presents a possible solution: expressible
sharing. Section 3 demonstrates expressible sharing on a small case-study, specifically the
description of a sequential odd-even transposition sorter. Section 5 discusses other solutions
to the sharing problem. Section 6 concludes.

2 The problem

In this section, we summarise a well-established method for describing synchronous circuits
in Haskell [2, 4]. We then demonstrate the problem which arises when trying to extract
circuit graphs from such descriptions.

2.1 Describing circuit behaviour

In a synchronous digital circuit, binary signals are processed at a rate dictated by a global
clock. In Haskell, such a signal can be modelled as an infinite list of boolean values, where
the n-th value in the list represents the value of the signal on the n-th clock cycle. We call
this type of signal a Bit using the following type synonym.

type Bit = [Bool]

The constant signals high and low are defined as follows.

high, low :: Bit
high = True : high
low = False : low

Bits can be processed using familiar logic gates, for example, an and-gate.

(<&>) :: Bit -> Bit -> Bit
a <&> b = zipWith (&&) a b

The output of an and-gate on the n-th clock-cycle is the boolean conjunction (&&) of its two
inputs on the n-th clock-cycle. Other logic gates can be defined in a similar way, for example
an xor-gate which combines signals using the disequality operator (/=).

(<#>) :: Bit -> Bit -> Bit
a <#> b = zipWith (/=) a b

We can also define components which have memory, for example, a component which
outputs an initial value on the first clock cycle, and thereafter outputs its input from the
previous clock cycle.

flipFlop :: Bit -> Bit -> Bit
flipFlop init a = head init : a

The design process is now one of combining primitive components like logic gates, flip-
flops and constants to yield more useful circuits. We have already seen the example circuit
flipper. Another example is a half-adder, shown in Fig. 3, which takes two bits, as inputs,
and returns the two-bit sum.

Fig. 3 Circuit diagram for
halfAdd

halfAdd :: Bit -> Bit -> (Bit, Bit)
halfAdd a b = (sum, carry)

where sum = a <#> b
carry = a <&> b

The circuits halfAdd and flipper can be simulated using a Haskell interpreter. (The
symbol > is used to represent the Haskell interpreter’s prompt.)

> halfAdd [False, True] [True, True]
([True, False], [False, True])

> flipper low
[True, False, True, False, True, False, ...

What we have essentially defined in this section is a data type Bit and a set of functions
for processing bits. In fact, Bit is an abstract data type (ADT): programmers should not
assume anything about its underlying representation and associated functions. The reason
for this will become clear in the next section where we change the implementation of the
Bit ADT so that we can extract other useful information from a circuit description.

2.2 Describing circuit structure

One particularly useful piece of information about a circuit is a list of all its components and
associated connections. This list, often referred to as a netlist, can for example be used to
realise a circuit by passing it to an external synthesis tool, or to verify properties about the
circuit by passing it to an external model checker [1].

To extract a netlist from a circuit description, we interpret Bit and its associated func-
tions symbolically. In other words, we supply variable names as input to the circuit, and
construct a data structure which represents the circuit. One symbolic interpretation of Bit
and its associated primitives is shown in Fig. 4.

If we symbolically evaluate halfAdd we obtain a data structure which correctly identi-
fies the presence of one and-gate and one xor-gate.

> halfAdd (Var "a") (Var "b")
(Xor2 (Var "a") (Var "b"), And2 (Var "a") (Var "b"))

Notice, however, that there are two occurrences of each input variable, because halfAdd
references each of its arguments twice. What if one of the inputs to halfAdd was not a
simple variable, but the output of, say, an and-gate?

> halfAdd (Var "a" <&> Var "b") (Var "c")
(Xor2 (And2 (Var "a") (Var "b")) (Var "c"),
And2 (And2 (Var "a") (Var "b")) (Var "c"))

Now we obtain a data structure which indicates the presence of three and-gates, despite the
fact that we only introduced one new and-gate to our description. We have lost the sharing
information which we intended in our description.

data Bit = High | Low | And2 Bit Bit
| Inv Bit | Var String | Xor2 Bit Bit | FF Bit Bit

high :: Bit
high = High

low :: Bit
low = Low

inv :: Bit -> Bit
inv a = Inv a

(<&>) :: Bit -> Bit -> Bit
a <&> b = And2 a b

(<#>) :: Bit -> Bit -> Bit
a <#> b = Xor2 a b

flipFlop :: Bit -> Bit -> Bit
flipFlop init a = FF init a

var :: String -> Bit
var s = Var s

Fig. 4 Symbolic implementation of the Bit ADT

The problem is compounded when we symbolically evaluate flipper, a circuit with a
feedback loop. The result is an infinitely large data structure.

> flipper (Var "init")
Inv (FF (Var "init") (Inv (FF (Var "init") (...

The problem is that when a variable is referenced more than once we can only determine
that each reference has the same value by comparing them for equality. For extremely large
values like netlists, comparing for equality may be infeasibly slow. And for cyclic data
structures, comparing for equality will not terminate.

3 Expressible sharing

The idea of expressible sharing is to make multiple references to a variable more explicit. To
do this, we introduce a new component called forkBit.

forkBit :: Bit -> (Bit, Bit)

Now if we want to use the value of a variable v twice, then we pass v to forkBit and obtain
two new variables, say v0 and v1, which can be used in place of original two references
to v. To illustrate, the definitions of flipper and halfAdd using expressible sharing are
shown in Fig. 5.

In the following sections we describe the details of expressible sharing, how it works,
and how it can be implemented.

3.1 General forking

To fork a single bit is easy, but to fork a large structure of bits one must deconstruct the
structure, extract and fork the individual bits, and finally put the forked bits into two new
structures. To simplify this task, we introduce a class of types whose values can be forked.

Fig. 5 Definitions of flipper
and halfAdd using expressible
sharing

flipper :: Bit -> Bit
flipper init = y0

where
x = inv (flipFlop init y1)
(y0, y1) = forkBit x

halfAdd :: Bit -> Bit -> (Bit, Bit)
halfAdd a b = (a0 <#> b0, a1 <&> b1)

where
(a0, a1) = forkBit a
(b0, b1) = forkBit b

class Fork a where
fork :: a -> (a, a)

The Bit type is an obvious member of the Fork class.

instance Fork Bit where
fork = forkBit

The attraction of type classes here is that they can be defined inductively. We can for example
express that if values of any type a can be forked, then so too can values of type list of a.

instance Fork a => Fork [a] where
fork = unzip . map fork

Similarly, if values of any types a and b can be forked, then so too can pairs with components
of type a and b.

instance (Fork a, Fork b) => Fork (a, b) where
fork (a, b) = ((a0, b0), (a1, b1))

where (a0, a1) = fork a
(b0, b1) = fork b

Together these instances define how to fork any structure containing any nested combination
of lists, pairs and bits.

3.2 How expressible sharing works

Assume for a moment that every value of type Bit contains a unique identifier called a
name. When a bit b is forked into two bits b0 and b1, it is clearly possible to mark the fact
that b0 and b1 refer to the same bit by storing the name of b in each. Hence, we can tell if
two bits are the same without comparing them for equality.

But how can we ensure that every value of type Bit contains a unique name? We reason
as follows.

– Input bits to the circuit can be given unique names by the programmer.
– Assuming that the inputs to a component are uniquely named, then the outputs can be

uniquely named too.
– Since bits are not arbitrarily copied, but only forked using forkBit, there is no scope

for a unique name to be duplicated.

However, the primitive components high and low do not take any inputs, and hence
have no source of a unique name. These constants are not supported by expressible shar-
ing; instead, the programmer must parameterise their circuits by any constant values used,
treating them like inputs.

Fig. 6 An example of a circuit
with no source of a unique name

type Bit = (Name, Expr)

name :: Bit -> Name
name (n, e) = n

expr :: Bit -> Expr
expr (n, e) = e

data Expr = Var Name | And2 Expr Expr | FF Expr Expr
| Inv Expr | Xor2 Expr Expr | Fork Name Expr

data Dir = L | R
type Name = [Dir]

var :: Name -> Bit
var n = (n, Var n)

inv :: Bit -> Bit
inv a = (name a, Inv (expr a))

(<&>) :: Bit -> Bit -> Bit
a <&> b = (name a, And2 (expr a) (expr b))

(<#>) :: Bit -> Bit -> Bit
a <#> b = (name a, Xor2 (expr a) (expr b))

flipFlop :: Bit -> Bit -> Bit
flipFlop a b = (name a, FF (expr a) (expr b))

forkBit :: Bit -> (Bit, Bit)
forkBit a = (b, c)
where f = Fork (name a) (expr a)

b = (L:name a, f)
c = (R:name a, f)

Fig. 7 Implementation of expressible sharing. Note the new component forkBit and the omission of the
constants high and low

Still, there remains a way to construct a circuit which has no source of a unique name:
simply connect the output of a component directly to its own inputs. For example, see Fig. 6.
This limitation will be discussed further in Sect. 3.4.

3.3 Implementation of expressible sharing

An implementation of expressible sharing is shown in Fig. 7. Like in Sect. 2.2, a Bit con-
tains a symbolic representation of a circuit. But now it also contains a unique name. The
components inv, <&>, <#>, and flipFlop still construct a symbolic representation, but
now also return a unique name, extracted from the first input bit of the component in each
case.

The most interesting part of the implementation is the new component forkBit. It
makes two copies of the symbolic representation of its input bit, but marks each copy with
a unique name using the Fork constructor. Any Fork constructor with the same name

represents the same circuit. The forkBit component also returns two new unique names
by prefixing the unique name in its input bit with values L and R respectively. Names are
represented as lists of such values precisely to allow easy forking of unique names.

Now consider evaluating halfAdd at a Haskell prompt. (Note that unique names of
equal length are supplied as inputs.)

> halfAdd (var [L,L] <&> var [L,R]) (var [R,L])
(([L,L,L], Xor2 (Fork [L,L] (And2 (Var [L,L]) (Var [L,R])))

(Fork [R,L] (Var [R,L]))),
([R,L,L], And2 (Fork [L,L] (And2 (Var [L,L]) (Var [L,R])))

(Fork [R,L] (Var [R,L]))))

Although we still obtain three and-gates, two of them are contained within a Fork construc-
tor with the same name [L,L]. Similarly, consider evaluating flipper and at a Haskell
prompt.

> flipper (var [L])
([L,L], Fork [L] (Inv (FF (Var [L]) (Fork [L] (Inv (FF (...

Although we still obtain an infinite data structure, we can detect the cycle by observing the
two Fork constructors with the same name.

3.4 Loop-restricted circuits

Section 3.2 showed that it is possible to construct circuits containing components which have
no source of a unique name (see Fig. 6 for an example). This section precisely identifies this
class of circuits, and discusses its significance.

Let us define a loop-restricted circuit to be one which is built out of the components
provided in Fig. 7 and in which every cyclic path contains at least one flipFlop whose
first input is not connected to a path originating from its own output. A netlist can be ex-
tracted, using expressible sharing, from any loop-restricted circuit because every cyclic path
is known to contain at least one flipFlop whose first argument is a source of a unique
name.

In contrast, a non loop-restricted circuit cannot be turned into netlist using expressible
sharing. But such a circuit has no meaning under the behavioural semantics of Sect. 2.1
because it is known to contain at least one black hole. A black hole is an unproductive cyclic
definition, for example let x = x in x.

In summary, expressible sharing does not make all kinds of loop observable. In particu-
lar, it does not make loops observable which do not depend on any external input. This is
because there is no source of a unique name, needed to identify the fork-points in cyclic
paths. However, in the circuit description library considered here, all meaningful loops are
observable.

3.5 Missing-fork detection

It is possible that a programmer may forget to use fork. Such a situation can be detected
because, when fork is not used where it should, the symbolic representation of the circuit
will contain:

1. more than two Fork constructors with the same name, or
2. more than one Var constructor with the same name.

Fig. 8 Diagram of a
four-element sequential odd-even
transposition sorter: a rectangle
containing an arrow represents a
two-sorter, which sorts two given
inputs; a trapezium represents a
multiplexor which selects one of
two inputs based on the value of
a condition bit; and a triangle
represents a register that outputs
its initial input on the first-clock
cycle and its second input,
delayed by one clock-cycle,
thereafter. In order of increasing
thickness, a line represents either
a single bit, a list of bits, or a list
of list of bits

We can check for these two conditions in a loop-restricted circuit by a top-down traversal
of the symbolic representation, during which we only visit the sub-circuit e of a value of the
form Fork n e once for any given name n. If either of these two conditions hold, then the
programmer must have copied a value of type Bit without using fork.

3.6 Copying versus forking

In some situations the programmer may wish to copy a circuit rather share its output. This
can be achieved by introducing a copyBit operator, defined like forkBit, but which
introduces a Copy constructor instead of a Fork. The child of a Copy constructor is then
replicated when the netlist is generated. This gives the programmer full control over what is
shared and what is copied.

4 A small case-study

In this section, we use a circuit description library based on expressible sharing to describe
a sequential odd-even transposition sorter [3]. The sorter is sequential in the sense that it
may take several clock-cycles to complete (linear with respect to the number of inputs). A
four-element instance of the sorter is shown in Fig. 8.

The sorter has two modes of operation. When mode is high, the values in inps are latched
into the internal registers of the sorter. When mode is low, the partially-sorted outputs of the
odd-even transposition logic are latched into the internal registers. A typical use-case of the
sorter is as follows: the user loads their own inputs into the sorter by setting mode to high,
and asserting inps appropriately. After setting mode back to low, the user waits � n

2 � clock
cycles by which stage the output of the sorter, outs, will be the fully-sorted inputs.

In Fig. 8, the sorter is divided into three separate stages using dotted lines: transpose,
multiplex, and delay. The next three sections define these stages individually, in reverse
order. After that, the top-level sorter definition is given.

4.1 The delay stage

The flipFlop component delays the value of a single bit by a clock-cycle. Using the fol-
lowing type-class, we generalise flipFlop so that whole structures of bits can be delayed.

class Delay a where
delay :: a -> a -> a

The type Bit is a member of the Delay type-class.

instance Delay Bit where
delay = flipFlop

Assuming that some type a can be delayed, then so too can a list of a.

instance Delay a => Delay [a] where
delay [] inp = []
delay (x:xs) inp = let y:ys = inp in delay x y : delay xs ys

So the delay function is capable of delaying a bit, a list of bits, a list of list of bits, and so
on.

4.2 The multiplex stage

The following circuit bitMux takes a condition bit and a pair of bits. If the condition is
high, the first bit of the pair is returned, otherwise the second bit of the pair is returned.

mux :: Bit -> (Bit, Bit) -> Bit
mux cond (a, b) = (cond0 <&> a) <|> (inv cond1 <&> b)

where (cond0, cond1) = fork cond

Note that the <|> operator represents an or-gate. Like above, we can generalise mux so that
the pair can contain not just single bits but whole structures of bits.

class Mux a where
(?) :: Bit -> (a, a) -> a

The type Bit is a member of the Mux type-class.

instance Mux Bit where
(?) = mux

Assuming that some type a can be multiplexed, then so too can a list of a.

instance Mux a => Mux [a] where
cond ? (a, b) = zipWith (?) (many cond) (zip a b)

The auxiliary function many forks its input an infinite number of times, making use of
Haskell’s non-strict semantics.

many :: Fork a => a -> [a]
many a = a0 : many a1

where (a0, a1) = fork a

Pairs whose components can be multiplexed can also be multiplexed.

instance (Mux a, Mux b) => Mux (a, b) where
cond ? ((a, b), (c, d)) = (cond0 ? (a, c), cond1 ? (b, d))

where (cond0, cond1) = fork cond

Together these instances define how to fork any structure containing any nested combination
of lists, pairs and bits.

Fig. 9 Circuit diagram for an
8-input instance of oet,
parameterised by a function f

Finally, the following function cond takes three functions c, f, and g as arguments. It
applies c to another argument a (which can be forked), and depending on the result, applies
f to a or g to a to give the result.

cond :: (Fork a, Mux b)
=> (a -> Bit) -> (a -> b) -> (a -> b) -> a -> b

cond c f g a = (c a0) ? (f a1, g a2)
where a0:a1:a2:_ = many a

4.3 The transpose stage

A key component in the sorter is the odd-even transposition pattern oet. In this pattern, each
pair of consecutive values in a list starting at an even index is passed through a function f,
then each pair starting at an odd index is passed through f. The function evenps performs
the former stage, and oddps the latter.

evenps :: ((a, a) -> (a, a)) -> [a] -> [a]
evenps f [] = []
evenps f [a] = [a]
evenps f (a:b:abs) = [c, d] ++ evenps f abs

where (c, d) = f (a, b)

oddps :: ((a, a) -> (a, a)) -> [a] -> [a]
oddps f [] = []
oddps f (x:xs) = [x] ++ evenps f xs

oet :: ((a, a) -> (a, a)) -> [a] -> [a]
oet f = oddps f . evenps f

See Fig. 9 for a diagram of an 8-input instance of oet.
In our application, we would like the function f to be a two-sorter, that is, a function

which sorts a pair of values. Furthermore, the values that we would like to sort are natural
numbers represented as a lists of bits, least significant bit first.

twoSort :: ([Bit], [Bit]) -> ([Bit], [Bit])
twoSort = cond gte swap id

where swap (a, b) = (b, a)

The function gte is a magnitude comparator which takes a pair of values and returns high
if the first value in the pair is greater than or equal to the second. Otherwise, gte returns
low.

Before defining gte, it is useful to define gteCell which takes a bit gteIn and a pair
of bits a and b. If a and b are not equal then a is returned as this represents whether a is
greater or less than b. If a and b are equal, then gteIn is returned.

gteCell :: Bit -> (Bit, Bit) -> Bit
gteCell gteIn (a, b) = gteOut

where eq = a0 <=> b
gteOut = eq ? (gteIn, a1)
(a0, a1) = fork a

The function gte can now be defined as a chain of gteCell. The initial value of gteIn
is high indicating that, initially, the inputs are considered to be equal.

gte :: ([Bit], [Bit]) -> Bit
gte (as, bs) = foldl gteCell high (zip as bs)

Unfortunately, this definition uses the constant high which is not supported by expressible
sharing. One solution would be to parameterise gte to take high as an input, but that would
require the functions which call gte to be modified. A simpler solution is just to partially
evaluate gte by hand and redefine it as follows.

gte (a:as, b:bs) = foldl gteCell (a <|> inv b) (zip as bs)

Notice that this version of gte works only on non-empty lists.

4.4 The sorter

One way to build a sorter is to compose � n
2 � copies of oet twoSort in sequence. Here,

we use a loop so that only one copy of oet twoSort is needed but consequently the sorter
requires � n

2 � clock cycles to complete. First, we introduce the function loop, akin to the
fix-point combinator.

loop :: Fork a => (a -> a) -> a
loop f = out0

where out = f out1
(out0, out1) = fork out

The sorter shown in Fig. 8 can now be defined as follows.

sort :: [[Bit]] -> Bit -> [[Bit]] -> [[Bit]]
sort inits mode inps =

loop (delay inits . (mode ?) . ((,) inps) . oet twoSort)

The init parameter is the list of numbers which the registers in the delay stage are ini-
tialised to. The inps parameter is the list of numbers which are loaded into the registers
when mode is high.

4.5 Summary

In the above case-study, the expression of fork-points was not a big burden. The unavail-
ability of the constants high and low was only a slight inconvenience in the definition of
gte as the use of a constant could be easily removed. However, in general one may need
to propagate constants from top-level inputs down to each point-of-use, and this may prove
somewhat more awkward.

5 Related work

5.1 Explicit naming

O’Donnell was the first to consider a solution to the netlist-generation problem [4]. His idea
was to assign a unique name to every instance of a component in the circuit description.
Sharing can then be detected by observing components with the same name. To illustrate,
here is the definition of halfAdd.

halfAdd :: Name -> Bit -> Bit -> (Bit, Bit)
halfAdd n a b = (sum, carry)

where sum = xor2 (n!0) (a, b)
carry = and2 (n!1) (a, b)

The main difference is that halfAdd now passes names to the primitive components it
uses. As halfAdd is a pure function it can’t invent unique names, so it too takes a name as
an argument. The function ! is used to create a new name out of an existing name.

There are two main problems with explicit naming. First, as pointed out by Claessen,
“there is nothing to stop the programmer from mistakenly introducing the same tag [name]
on different components” [1]. Second, names have to be manually threaded all the way from
the top-level function down to the low-level components.

O’Donnell has recently proposed Template Haskell as a vehicle for automating the
process of explicit naming [5]. The idea is that circuit descriptions can be automatically
parsed, explicitly named, and unparsed, all within the Template Haskell framework. The ad-
vantage of this approach is that the natural style of functional circuit description is preserved.
One disadvantage seems to be that it requires a lot of implementation effort. The approach
is impure because, although descriptions are written in a Haskell syntax, the meanings of
the programs are changed by the transformation.

5.2 Monadic naming

Having to explicitly pass around extra arguments and use them in same way again and again
is a classic problem with pure functional programming.

Pure functional languages have this advantage: all flow of data is made explicit. And
this disadvantage: sometimes it is painfully explicit. (Wadler [7])

Wadler proposed that a monad can be used to allow data to flow implicitly in a pure func-
tional program. Accordingly, Claessen devised a monad with the ability to perform implicit
naming of components [1]. To illustrate, here is the definition of halfAdd. (Claessen’s
monad is named LavaM.)

halfAdd :: Bit -> Bit -> LavaM (Bit, Bit)
halfAdd a b = do sum <- xor2 (a, b)

carry <- and2 (a, b)
return (sum, carry)

One of the disadvantages of the monadic approach is that we lose the ability to connect
circuits together simply by writing nested applicative expressions. Instead, expressions must
be flattened, expressing the order of evaluation, even though circuits are naturally thought
of as executing in parallel. Claessen writes: “what we are looking for is a solution that does
not require a change in the natural circuit description style, but allows us to detect sharing
and loops in a description from within the language” [1].

5.3 Observable sharing

The treatment of expressions as graphs is a standard technique in the implementation of
functional languages. Claessen and Sands exploit this fact by extending Haskell with a fea-
ture called observable sharing whereby references to data constructors can be compared for
equality [2]. The advantage is that programmers can express sharing simply by binding a
variable to an expression and referencing it multiple times. In addition, observable sharing
is easy to implement in several existing Haskell compilers. One disadvantage is that the
following property about Haskell programs is lost.

let {x = M} in N = N [M/x] (x /∈ M)

The authors argue that the loss of this property is not so bad, and that although compilers
could rely on its presence, many common compiler transformations are unaffected. In our
experience, observable sharing works well in all the main Haskell implementations at the
time of writing. Nevertheless, it is a non-conservative extension which is perhaps unlikely
to be adopted by standard Haskell.

6 Conclusion

We have explored the design of a circuit description library for a pure functional language
in which the programmer expresses sharing by being explicit about the fork-points in the
circuit. Circuit descriptions written in this way can be turned into netlists in which the ex-
pressed sharing is observable within the pure functional library. However, there are some
restrictions. The constant components high and low are not supported and must removed
by partial evaluation or taken as inputs and explicitly propagated to the point of use. Ex-
pressible sharing also requires that each loop depends on some external input, although this
restriction is not problematic in the particular circuit description library considered here.

